J. Fac. Fish. Anim. Husb.,
Hiroshima Univ. (1976), **15** : 1 ~ 15

Studies of Drying-rate Equations based on Drying-shell Models

Kiyoshi KUBOTA, Kanichi SUZUKI, Hideaki HOSAKA,
Reiko HIROTA and Kiyoshi IHARA

*Department of Food Chemistry and Technology,
Faculty of Fisheries and Animal Husbandry,
Hiroshima University, Fukuyama*

Received February 2, 1976

(Figs. 1 ~ 5, Appendix)

INTRODUCTION

Drying operations of foods are among the researches which have attracted special attention recently. The reason is that they help the improvement of food storage and transportation, which are crucial problems and in the food crisis that the world is going to account with in the near future. In order to design and automatically control various drying apparatuses, it is required to determine the drying-rate equations and to obtain the rate parameters for the equations.

The basic studies of the drying-rate equations have been investigated[1),2)] dividing the drying period into a constant- and a falling-rate periods. However, the drying phenomena of foods that have cell tissues and have many components, are complicated, and the drying-rate curves do not always keep a constant shape, so it becomes difficult to apply the methods descrived above, to the drying-rate equations. As an urgent question, for the requirements in the design of various drying apparatuses, we must take the drying-rate equations as approximating equations based on the some simple drying models, and we must simulate the rate parameters in the rate equations from the experimental data.

In this study, we postulate the integrated equations of the drying-rate equations based on the drying-shell models which considered the surface-shrinkage for the sphere, the long-cylinder and the infinite-slab, respectively. Then, we report on the calculation methods of two rate parameters in the drying-rate equations using a non-linear least square method.

DRYING-RATE EQUATIONS

Drying mechanisms of foods are divided into two approximated types. One type has two simple idealized zones, the undrying-core and the drying-shell. Thus, the moisture content of the undrying-core is assumed as the initial moisture content and the drying-shell is taken for the equilibrium moisture content. The other type has the distribution of moisture content throughout the whole of the material, and the moisture content decreases gradually toward the surface. The former type (drying-shell model) is used under accelerated drying conditions such as high air velocity, high temperature and low humidity, and the latter type (drying-whole model) is used under loosen conditions. The former shell type models for the sphere, the long-cylinder and the infinite-slab for catalytic reactions have been reported[3),4)], but have not yet been used for the investigation of drying-rate equations. The integrated equations of the rate equations for obtaining the rate parameters of shell diffusion which considering of the shrinkage of surface, have not yet been obtained. In this paper, we determine the drying-rate equations on the basis of the former drying-shell models, and we induce the integrated equations.

1. Drying-rate equations for sphere

Fig. 1 illustrates the case of a drying spherical material. The radius, the volume and the weight of the spherical material in the intermediate state, are given as R(cm), V(cm^3) and w(g), and the radius of the undrying-core is givens as r_c(cm). The subscripts o, e and d in these symbols signify the initial, equilibrium and completely drying states, respectively.



Fig. 1. Drying-shell model for sphere.

Thus, we may define the moisture content W(g-H_2O/g-D.M.) and the drying-ratio X_W(-) of material by the following equations.

$$W = (w - w_d)/w_d \tag{1}$$

$$X_w = (W_o - W)/(W_o - W_e) = (w_o - w)/(w_o - w_e) \tag{2}$$

In the intermediate state, if the drying-shell and undrying-core states are the same the equilibrium and initial states, respectively, the relations of the radius of undrying-core r_c, the drying-ratio X_w and the radius of spherical material R become as follows:

$$r_c = (1 - X_w)^{1/3} R_o \tag{3}$$

$$R = R_e{}^3 + (1 - (R_e/R_o)^3)r_c{}^3)^{1/3} \tag{4}$$

The material balance equation of drying rate is obtained, for the decreasing of a very small radius of undrying-core dr_c(cm), in a very short time interval $d\theta$(min), where $\mathcal{R}$(g-H_2O/min) is the drying-rate of material.

$$dr_c/d\theta = \mathcal{R}/(4\pi r_c{}^2 \rho_h) \tag{5}$$

where, $\rho_h = (w_o - w_e)/V_o$ (6)

For the diffusion controllings, when the moisture concentrations of the undrying-core, drying-shell and gas-film surfaces are c_s, c_i and c_g (g-H_2O/cm^3-void), respectively, the drying-rate $\mathcal{R}$ can be written as follows:

For gas-film diffusion controlling:

$$\mathcal{R} = 4\pi R^2 h_m(c_i - c_g) \tag{7}$$

where, h_m: rate parameter of gas-film diffusion (cm^3-void/cm^2·min)

For shell diffusion controlling:

$$\mathcal{R} = (4\pi R r_c/(R - r_c))k_m(c_s - c_i) \tag{8}$$

where, k_m: rate parameter of shell diffusion (cm^3-void/cm·min)

The drying-rate equation for diffusion controlling is obtained by combining Eqs. (7) and (8), and eliminating c_i.

$$\mathcal{R} = 4\pi R^2 (c_s - c_g)/((1/h_m) + ((R - r_c)/((r_c/R)k_m))) \tag{9}$$

For the heat-transfer controllings, when the temperatures of the undrying-core, drying-shell and gas-film surfaces are t_s, t_i and t_g (°C), respectively, the drying-rate $\mathcal{R}$ can be written as follows:

For gas-film heat-transfer controlling:

$$\mathcal{R} = (4\pi R^2/(\Delta H))h_h(t_g - t_i) \tag{10}$$

where, h_h: rate parameter of gas-film heat-transfer (cal/cm^2·min·°C), ΔH: latent heat of vaporization (cal/g-H_2O)

For shell heat-transfer controlling:

$$\mathcal{R} = (4\pi R r_c/((R - r_c)(\Delta H)))\, k_h(t_i - t_s) \tag{11}$$

where, k_h: rate parameter of shell heat-transfer (cal/cm·min·°C)

The drying-rate equation for heat-transfer controlling is obtained by combining Eqs. (10) and (11), and eliminating t_i.

$$\mathcal{R} = 4\pi R^2 (t_g - t_s)/(((\Delta H/h_h) + ((R - r_c)(\Delta H)/(r_c/R)k_h)) \tag{12}$$

Thus, combining Eqs. (9) and (12), we obtain the following relation, then the rate parameters of diffusion and heat-transfer controllings can be converted into each other.

$$h_m(c_s - c_g) = h_h(t_g - t_s)/(\Delta H) \tag{13}$$

$$k_m(c_s - c_g) = k_h(t_g - t_s)/(\Delta H) \tag{14}$$

2. Drying-rate equations for long-cylinder



Fig. 2. Drying-shell model for long-cylinder.

Fig. 2 illustrates the situation of a drying long-cylinder, hence the length L(cm) is very much longer than the radius R(cm), and the radius of undrying-core givens r_c(cm).

Thus, analogically to the sphere, the relations of the radius of undrying-core r_c, the drying-ratio X_w and the radius of long cylindrical material R become as follows:

$$r_c = (1 - X_w)^{1/2} R_o \tag{15}$$

$$R = (R_e^2 + (1 - (R_e/R_o)^2) r_c^2)^{1/2} \tag{16}$$

The material balance equation is obtained analogous to the sphere too.

$$dr_c/d\theta = -\mathcal{R}/(2\pi r_c L \rho_h) \tag{17}$$

For the diffusion controlling, the drying-rate $\mathcal{R}$ can be written as follows:

$$\mathcal{R} = 2\pi R L h_m(c_i - c_g) \tag{18}$$

$$\mathcal{R} = (2\pi L/\ln(R/r_c)) k_m(c_s - c_i) \tag{19}$$

The drying-rate equation for diffusion controlling is obtained by combining Eqs. (18) and (19).

$$\mathcal{R} = 2\pi R L(c_s - c_g)/((1/h_m) + (R\ln(R/r_c)/k_m)) \tag{20}$$

For the heat-transfer controlling, the drying-rate equation is obtained from Eq. (20) on combining Eqs. (13) and (14).

3. Drying-rate equations for infinite-slab

Fig. 3 illustrates the situation of a drying infinite-slab, hence the width and the longitudinal length are very much longer than the half-thickness L(cm), and the half-thickness of undrying-core and the one side area given x_c(cm) and A(cm^2), respectively.

Thus, analogically to the sphere, the relations of the half-thickness of undrying-core x_c, the drying-ratio X_w and the half-thickness of infinite-slab's material L become as follows:

$$x_c = (1 - X_w)L_o \tag{21}$$



Fig. 3. Drying-shell model for infinite-slab.

$$L = L_e + (1 - (L_e/L_o))x_c \tag{22}$$

The material balance equation is obtained analogous to the sphere too.

$$dx_c/d\theta = -\mathcal{R}/(2A\rho_h) \tag{23}$$

For the diffusion controlling, the drying-rate $\mathcal{R}$ can be written as follows:

$$\mathcal{R} = 2Ah_m(c_i - c_g) \tag{24}$$

$$\mathcal{R} = (2A/(L - x_c))\,k_m(c_s - c_i) \tag{25}$$

The drying-rate equation for diffusion controlling is obtained by combining Eqs. (24) and (25).

$$\mathcal{R} = 2A(c_s - c_g)/((1/h_m) + ((L - x_c)/k_m)) \tag{26}$$

For the heat-transfer controlling, the drying-rate equation is obtained from Eq. (26) on combining Eqs. (13) and (14).

CALCULATION METHODS OF RATE PARAMETERS

1. Gas-film or shell diffusion controlling

For the cases of only a partly controlling in the diffusion and heat-transfer controllings, we could integrate the drying-rate equations, and the rate parameters can be calculated in explicit function.

For the rate parameter h_m of spherical material, the integrated equation can be obtained by substituting Eqs. (4) and (7) into Eq. (5), and integrating $\theta = 0 \to \theta$, $r_c = R_o \to r_c$, hence $c_i = c_s$. For the rate parameter k_m, the equation can be obtained by substituting Eqs. (4) and (8) into Eq. (5), and integrating, hence $c_i = c_g$. Then, the rate parameters h_h and k_h can be obtained from Eqs. (13) and (14) using the values of h_m and k_m, respectively.

Thus, let us illustrate the integrated euqations for the diffusion controllings.

(1) Spherical material

For shrinking surface:

$$h_m = \rho_h R_o^2 (R_e^2 - R(R_o^3 R_e^3 + r_c^3 (R_o^3 - R_e^3))^{1/3}) /((R_o^3 - R_e^3)(c_s - c_g)\theta) \tag{27}$$

$$k_m = \rho_h (R_o (R_o^3 R_e^3 + r_c^3 (R_o^3 - R_e^3))^{2/3} - (R_o^2 R_e^3 + r_c^2 (R_o^3 - R_e^3))) /(2(R_o^3 - R_e^3)(c_s - c_g)\theta) \tag{28}$$

For constant surface:

$$\begin{aligned} h_m &= \rho_h (R_o^3 - r_c^3)/(3R_o^2 (c_s - c_g)\theta) \\ &= \rho_h R_o X_w /(3(c_s - c_g)\theta) \end{aligned} \tag{29}$$

$$\begin{aligned} k_m &= \rho_h (R_o^3 + 2r_c^3 - 3R_o r_c^2)/(6R_o (c_s - c_g)\theta) \\ &= \rho_h R_o^2 (1 - 3(1 - X_w)^{2/3} + 2(1 - X_w))/(6(c_s - c_g)\theta) \end{aligned} \tag{30}$$

(2) Long-cylindrical material

For shrinking surface:

$$h_m = \rho_h R_o (R_o^2 - (R_o^2 R_e^2 + r_c^2 (R_o^2 - R_e^2))^{1/2})/((R_o^2 - R_e^2)(c_s - c_g)\theta) \tag{31}$$

$$k_m = \rho_h (2r_c^2 (R_o^2 - R_e^2)\ln r_c + (4R_o^2 R_e^2 + 2r_c^2 (R_o^2 - R_e^2))\ln R_o - (R_o^2 R_e^2 + r_c^2 (R_o^2 - R_e^2))\ln(R_o^2 R_e^2 + r_c^2 (R_o^2 - R_e^2)))/(4(R_o^2 - R_e^2)(c_s - c_g)\theta) \tag{32}$$

For constant surface:

$$\begin{aligned} h_m &= \rho_h (R_o^2 - r_c^2)/(2R_o (c_s - c_g)\theta) \\ &= \rho_h (R_o X_w /(2(c_s - c_g)\theta) \end{aligned} \tag{33}$$

$$\begin{aligned} k_m &= \rho_h (R_o^2 - r_c^2 + 2r_c^2 \ln r_c - 2r_c^2 \ln R_o)/(4(c_s - c_g)\theta) \\ &= \rho_h R_o^2 (X_w + (1 - X_w)\ln(1 - X_w))/(4(c_s - c_g)\theta) \end{aligned} \tag{34}$$

(3) Infinite-slab's material

For shrinking surface:

$$\begin{aligned} h_m &= \rho_h (L_o - x_c)/((c_s - c_g)\theta) \\ &= \rho_h L_o X_w /((c_s - c_g)\theta) \end{aligned} \tag{35}$$

$$\begin{aligned} k_m &= \rho_h L_e (L_o - x_c)^2 /(2L_o (c_s - c_g)\theta) \\ &= \rho_h L_o L_e X_w^2 /(2(c_s - c_g)\theta) \end{aligned} \tag{36}$$

For constant surface:

h_m = same to Eq. (35)

$$\begin{aligned} k_m &= \rho_h (L_o - x_c)^2 /(2(c_s - c_g)\theta) \\ &= \rho_h L_o^2 X_w^2 /(2(c_s - c_g)\theta) \end{aligned} \tag{37}$$

2. Gas-film and shell diffusion controllings

For the case of two or more over part controllings in the diffusion and heat-transfer controllings, the drying-rate equation can't be integrated analytically so it must be integrated numerically. The experimental data are generally given the integrated type as weight of material w vs. drying time θ. For the calculations of the rate parameters h_m and k_m of spherical material, we substitute Eqs. (4) and (9) into Eq. (5), and integrate numerically $\theta = 0 \rightarrow \theta$; $r_c = R_o \rightarrow r_c$. Using a Runge-Kutta-Gill method and so on, the differential type $dr_c/d\theta$ in Eq. (5) are converted to r_c.

The following standard deviation σ(cm) for r_c is minimized.

$$\sigma = \left(\sum_{i=1}^{n} (r_{c,obs} - r_{c,cal})_i^2 \, W_i/n\right)^{1/2} \tag{38}$$

where, W_i: weighting coefficient for $(r_c)_i$

($i = 1 \sim$ n: number of experimental points)

The rate parameters are non-linear, and then must be calculated by a non-linear least square method using an electronic computer. The calculations using a non-linear least square method is easily performed by using a previously established subroutine program[5),6)] for a digital computer.

The equations of a non-linear least square method are shown as the l element linear simultaneous equations, and we calculate the correction values Δa_p of the initial parameters $a_{p,o}$ from the equations. The corrected values of parameters $a_{p,o} + \Delta a_p$ are used as the initial values of second iteration and so on.

$$(\mathbf{A} + \lambda \mathbf{D})\delta = \nu \mathrm{g} \tag{39}$$

where,

$\mathbf{A} = [\sum_{i=1}^{n} ((\partial r_c/\partial a_p)_o(\partial r_c/\partial a_{p'})_o)_i W_i^2]$: (l, l) matrix

$\mathbf{D}$: diagonal matrix of $\mathbf{A}$

$\boldsymbol{\delta} = [a_p - a_{p,o}] = [\Delta a_p]$: (l) column vector

$\mathbf{g} = [\sum_{i=1}^{n} ((r_{c,obs} - r_{c,o})\,(\partial r_c/\partial a_p)_o)_i W_i^2]$: (l) column vector

λ : weighting factor between Gauss method ($\lambda = 0$) and steepest descent method ($\lambda = \infty$), ν : step size, a_p : parameter ($p = 1 \sim l$: number of parameters)

The flow chart of a digital computer for the calculation of the rate parameters in the drying-rate equations is shown in Fig. 4. The example of the flow chart of the main program for the spherical material at constant surface is shown in Fig. 5, and the practical program is shown in Appendix (used of the Computation Center of Hiroshima University: TOSBAC 3400-14).

The subroutine program of a non-linear least square method HISENS in Appendix may be used widely in various researches, and the methods to apply the practical program



Fig. 4. Flow chart for calculation of non-linear rate parameters.



Fig. 5. Flow chart of main program.

were previously reported[5),6)]. The values of X_W, w, W and $dW/d\theta$ compared to the observed values respectively, are calculated from r_c and $dr_c/d\theta$ using the following equations.

For spherical material:

$$X_W = 1 - (r_c/R_o)^3 \quad \text{from Eq. (3)} \tag{40}$$

$$w = w_o(1 - X_W) + w_e X_W \quad \text{from Eq. (2)} \tag{41}$$

W = same to Eq. (1)

$$dW/d\theta = (3(1 - X_W)^{2/3}(w_O - w_e)/(R_O w_d))dr_c/d\theta \quad (42)$$

from differentiate Eqs. (2) and (3) with respect to θ

For long-cylindrical material:

$$X_W = 1 - (r_c/R_O)^2 \quad \text{from Eq. (15)} \quad (43)$$

w =, W = same to Eqs. (41) and (1)

$$dW/d\theta = (2(1 - X_W)^{1/2}(w_O - w_e)/(R_O w_d))dr_c/d\theta \quad \text{from Eqs. (12) and (15)} \quad (44)$$

For infinite-slab's material:

$$X_W = 1 - (x_c/L_O) \quad \text{from Eq. (21)} \quad (45)$$

w =, W = same to Eqs. (41) and (1)

$$dW/d\theta = ((w_O - w_e)/(L_O w_d))dx_c d\theta \quad \text{from Eqs. (2) and (21)} \quad (46)$$

Appendix and Fig. 5 show the calculation for the calculated values of w as an example of these calculations.

RESULTS

The drying-rate equations based on the drying-shell models were postulated, and then the calculation methods of the non-linear rate parameters were reported.

The drying-rate equations in this paper are characterized by the convenience when the constant- and falling-rate periods are not dearly observed and the shrinkage of the surface becomes evident, and are used commonly in drying of foods.

SUMMARY

In order to design and automatically control various drying apparatuses, it is required to determine the drying-rate equations and to obtain the rate parameters for the equations. In this study, we postulated the integrated equations of the drying-rate equations based on the drying-shell models which considered the surface-shrinkage for the sphere, the long-cylinder and the infinite-slab, respectively. Then, we reported on the calculation methods of the rate parameters using a non-linear least square method.

The drying-rate equations in this paper are characterized by the convenience when the constant- and falling-rate periods are not dearly observed and the shrinkage of the surface becomes evident, and are used commonly in drying of foods.

NOTATIONS

A : one side area of infinite-slab (cm^2)
c : moisture concentration ($g\text{-}H_2O/cm^3$-void)
ΔH : latent heat of vaporization ($cal/g\text{-}H_2O$)
h_m : rate parameter of gas-film diffusion (cm^3-void/$cm^2 \cdot min$)
h_h : rate parameter of gas-film heat-transfer ($cal/cm^2 \cdot min \cdot °C$)
k_m : rate parameter of shell diffusion (cm^3-void/$cm \cdot min$)
k_h : rate parameter of shell heat-transfer ($cal/cm \cdot min \cdot °C$)
L : length of long-cylinder or half-thickness of inifinite-slab (cm)
R : radius of sphere or long-cylinder (cm)
r_c : radius of undrying-core of sphere or long-cylinder (cm)
$\mathcal{R}$: drying-rate ($g\text{-}H_2O/min$)
V : volume of drying material (cm^3)
W : moisture content ($g\text{-}H_2O/g$-D.M.)
w : weight of drying material (g)
$dW/d\theta$: drying-rate ($g\text{-}H_2O/min \cdot g$-D.M.)
X_w : drying-ratio (–)
x_c : half-thickness of undrying-core of infinite-slab (cm)
θ : drying time (min)
ρ_h : water concentration of vaporization ($g\text{-}H_2O/cm^3$)

Subscripts;

o: initial state, e: equilibrium state, d: completely drying state,
g: gas-film surface, i: drying-shell surface, s: undrying-core surface

REFERENCES

1) SARAVACOS, G. D. and CHARM, S. E.: *Food Tech.*, **16**(1), 78–81 (1962).
2) PERRY, J. H.: Chemical Engineer's Handbook 4-ed., pp.(15) 32–43, McGraw-Hill Book Comp., New York (1963).
3) LEVENSPIEL, O.: Chemical Reaction Engineering, pp. 338–357, John Wily and Sons, Inc., New York (1962).
4) KUBOTA, H.: Kaisetsu Hannou Sousa Sekkei, pp. 195–205, Nikkan Kogyo Shinbunsha, Tokyo (1970).
5) KUBOTA, K. and MORITA, N.: *Computation Center News of Nagoya University,* **4**(4), 318–326 (1973). (in Japanese).
6) HOSAKA, H., KUBOTA, K. and SUZUKI, K.: Shokuhin Kogaku, pp. 170–179, Kyoritsu Shuppan Co., Ltd., Tokyo (1975).

APPENDIX

```
C       MAIN PROGRAM
C       CALCULATION OF RATE PARAMETERS IN DRYING-RATE EQUATION OF DRYING-
C       SHELL MODEL (SQUARE, CONSTANT SURFACE)
C       NON LINEAR LEAST SQUAR METHOD (HISENS),
C       FROM H.HOSAKA,K.KUBOTA AND K.SUZUKI, SHOKUHIN KOGAKU, PP.170,
C       KYORITSU SHUPPAN(1975)
C       N= TOTAL NUMBER OF EXPERIMENTAL POINT
C       T(N)= TIME, W(N)=WEIGHT, RC(N)= RADIUS OF UNDRYING-CORE
C       AH(N),AK(N)= RATE PARAMETERS OF GAS-FILM AND SHELL DIFFUSION
C       FROM INTEGRATED EQUATIONS
C       K1,K= NUMBER OF INITIAL SETTING AND ITERATION ENDING POINTS
C       A(1,K),A(2,K)=RATE PARAMETERS OF GAS-FILM AND SHELL DIFFUSION,
C       FROM NON-LINEAR LEAST SQUARE METHOD
C       SD(K)= STANDARD DEVIATION
C       YC(1,N,K)= CALCULATION VALUES OF RC(N)
C       WC(N,K)= CALCULATION VALUES OF W(N)
        DIMENSION T(70),W(70),TS(70),WS(70),X(70),RC(70),AH(70),AK(70),
       1YY(1,70),XX(1,70),A(2,16),HM(16),HS(16),WW(1,70),YC(1,70,16),
       2YCS(1,70,3),DD(2,3),SD(16),XC(70,16),WC(70,16)
        COMMON RO,RE,CS,CG,RH
        DATA HAA/0.0005/,HM(1),HS(1)/2*1.0/,HMM,HSS/2*0.5/,HST/0.005/,
       1EPS/1.0E-40/,MY,MX/2*1/,L/2/,NT/70/,KT/15/
        EXTERNAL SUB
    777 READ(5,10) N
        IF(N) 999,999,888
    888 READ(5,20) RO,RE,WO,WE,WD,CS,CG,RH
        READ(5,30) (T(I),I=1,N)
        READ(5,30) (W(I),I=1,N)
C       EXPERIMENTAL VALUES OF 5 POINTS AVERAGE METHOD
        DO 100 I=1,N
        IF(I-2) 110,120,130
    110 II=I+2
        GO TO 170
    120 II=I+1
        GO TO 170
    130 N1=N-1
        IF(I-N1) 140,150,160
    140 II=I
        GO TO 170
    150 II=I-1
        GO TO 170
    160 II=I-2
    170 WS(I)=(W(II+2)+W(II+1)+W(II)+W(II-1)+W(II-2))/5.0
        IF(I-3) 100,190,180
    180 N2=N-2
        IF(N2-I) 100,190,190
    190 I2=I-2
        TS(I2)=T(I)
    100 WS(I2)=WS(I)
        N=N-4
        DO 200 I=1,N
        T(I)=TS(I)
        W(I)=WS(I)
    200 X(I)=(WO-W(I))/(WO-WE)
        WRITE(6,50) N,RO,RE,WO,WE,WD,CS,CG,RH
        WRITE(6,60) (T(I),I=1,N)
        WRITE(6,70) (W(I),I=1,N)
C       RATE PARAMETERS AH(N),AK(N)= FROM INTEGRATED EQUATIONS
        DO 300 I=1,N
        AX=1.0-X(I)
        RC(I)=AX**(1.0/3.0)*RO
        AH(I)=RH*RO*X(I)/(3.0*(CS-CG)*T(I))
    300 AK(I)=RH*RO**2*(1.0-3.0*AX**(2.0/3.0)+2.0*AX)/(6.0*(CS-CG)*T(I))
        WRITE(6,80) (AH(I),I=1,N)
        WRITE(6,85) (AK(I),I=1,N)
C       RATE PARAMETERS A(1,K),A(2,K)= FROM NON-LINEAR LEAST SQUARE METHOD
        L1=L+1
        K1=KT+1
        A(1,1)=AH(1)
        A(2,1)=AK(N)
        DO 400 I=1,N
```

```
      YY(1,I)=RC(I)
      XX(1,I)=T(I)
  400 WW(1,I)=1.0
      CALL HISENS(YY,XX,A,HAA,HM,HMM,HS,HSS,HST,WW,YC,YCS,DD,SD,
     1MY,MX,N,NT,L,L1,K,KT,K1,SUB,EPS,ILL)
      IF(ILL) 410,420,410
  410 WRITE(6,90) ILL
      GO TO 999
  420 WRITE(6,95) (A(IA,K1),IA=1,L),SD(K1)
      WRITE(6,96) K,(A(IA,K),IA=1,L),SD(K)
C     CALCULATED VALUES FROM OBTAINED RATE PARAMETERS
      DO 500 I=1,N
      XC(I,K1)=1.0-(YC(1,I,K1)/R0)**3
      WC(I,K1)=W0*(1.0-XC(I,K1))+WE*XC(I,K1)
      DO 500 IK=1,K
      XC(I,IK)=1.0-(YC(1,I,IK)/R0)**3
  500 WC(I,IK)=W0*(1.0-XC(I,IK))+WE*XC(I,IK)
      WRITE(6,97) (WC(I,K1),I=1,N)
      WRITE(6,98) (WC(I,K),I=1,N)
      GO TO 777
  999 STOP
   10 FORMAT(I4)
   20 FORMAT(5F8.0,3F12.0)
   30 FORMAT(10F8.0)
   50 FORMAT(1H1,26HN,R0,RE,W0,WE,WD,CS,CG,RH=/1H ,I4,4X,8E13.5)
   60 FORMAT(1H0,5HT(N)=/(1H ,10E13.5))
   70 FORMAT(1H0,5HW(N)=/(1H ,10E13.5))
   80 FORMAT(1H0,6HAH(N)=/(1H ,10E13.5))
   85 FORMAT(1H0,6HAK(N)=/(1H ,10E13.5))
   90 FORMAT(1H0,4HILL=,I8)
   95 FORMAT(1H0,15HA(L,K1),SD(K1)=/(1H ,10E13.5))
   96 FORMAT(1H0,16HK=,A(L,K),SD(K)=/1H ,I4,4X,3E13.5)
   97 FORMAT(1H0,9HWC(N,K1)=/(1H ,10E13.5))
   98 FORMAT(1H0,8HWC(N,K)=/(1H ,10E13.5))
      END
      SUBROUTINE HISENS(Y,X,A,HAA,HM,HMM,HS,HSS,HST,W
     1,YC,YCS,DD,SD,MY,MX,N,NT,L,L1,K,KT,K1,SUB,EPS,ILL)
C     NON-LINEAR LEAST SQUARE METHOD
C     H.HOSAKA,K.KUBOTA AND K.SUZUKI, SHOKUHIN KOGAKU, PP.170,
C     KYORITSU SHUPPAN(1975)
      DIMENSION Y(MY,NT),X(MX,NT),A(L,K1),HM(K1),HS(K1),W(MY,NT)
     1,YC(MY,NT,K1),YCS(MY,NT,L1),DD(L,L1),SD(K1),HA(20)
     2,AS(20),DDS(20)
      IF(KT.GE.1.OR.MY.GE.1.OR.MX.GE.1.OR.N.GE.L.OR.L.GE.2.OR
     1.L.LE.20.OR.L.LT.L1.OR.KT.LT.K1.OR.EPS.GE.0.0) GO TO 200
      ILL=30000
      GO TO 999
  200 K=1
      HM(K1)=HM(1)
      HS(K1)=HS(1)
      SDS=0.0
      DO 10 IA=1,L
   10 A(IA,K1)=A(IA,1)
  888 CALL SUB(YC,X,A,MY,MX,N,NT,L,K,K1,ILL)
      IF(ILL) 300,400,300
  300 IF(SDS) 999,999,700
  400 D=0.0
      DO 20 I=1,N
      DO 20 J=1,MY
   20 D=D+((Y(J,I)-YC(J,I,K))*W(J,I))**2
      SD(K)=SQRT(D/(N*MY))
      IF(SDS) 999,500,600
  500 SDS=SD(K)
      SD(K1)=SD(K)
      DO 30 I=1,N
      DO 30 J=1,MY
      YCS(J,I,L1)=YC(J,I,K)
   30 YC(J,I,K1)=YC(J,I,K)
      DO 21 IA=1,L
      A(IA,K1)=A(IA,1)
   21 CONTINUE
      GO TO 777
  600 IF(SDS-SD(K)) 700,800,800
  700 HS(K)=HS(K)*HSS
      IF(HST-HS(K)) 666,999,999
```

```
  800 IF(KT-K) 999,999,900
  900 K=K+1
      HM(K)=HM(K-1)*HMM
      HS(K)=HS(1)
      SDS=SD(K-1)
      DO 40 IA=1,L
   40 A(IA,K)=A(IA,K-1)
      DO 50 I=1,N
      DO 50 J=1,MY
   50 YCS(J,I,L1)=YC(J,I,K-1)
  777 DO 60 IA=1,L
      AST=A(IA,K)
      HA(IA)=A(IA,K)*HAA
      A(IA,K)=A(IA,K)+HA(IA)
      CALL SUB(YC,X,A,MY,MX,N,NT,L,K,K1,ILL)
      IF(ILL) 999,1000,999
 1000 DO 70 I=1,N
      DO 70 J=1,MY
      YCS(J,I,IA)=YC(J,I,K)
   70 YCS(J,I,IA)=(YCS(J,I,IA)-YCS(J,I,L1))/HA(IA)
   60 A(IA,K)=AST
      DO 80 IA1=1,L
      DO 80 IA2=1,L
      DD(IA1,IA2)=0.0
      DO 80 I=1,N
      DO 80 J=1,MY
   80 DD(IA1,IA2)=DD(IA1,IA2)+YCS(J,I,IA1)*YCS(J,I,IA2)
     1*W(J,I)**2
      DO 90 IA=1,L
      DD(IA,IA)=DD(IA,IA)*(1.0+HM(K))
      DD(IA,L1)=0.0
      DO 90 I=1,N
      DO 90 J=1,MY
   90 DD(IA,L1)=DD(IA,L1)+YCS(J,I,IA)*(Y(J,I)-YCS(J,I,L1))
     1*W(J,I)**2
      CALL GAUYOS(DD,L,L1,EPS,ILL)
      IF(ILL) 999,1100,999
 1100 DO 110 IA=1,L
      AS(IA)=A(IA,K)
  110 DDS(IA)=DD(IA,L1)
  666 DO 120 IA=1,L
      DD(IA,L1)=DDS(IA)*HS(K)
  120 A(IA,K)=AS(IA)+DD(IA,L1)
      GO TO 888
  999 RETURN
      END


      SUBROUTINE SUB(YC,X,A,MY,MX,N,NT,L,K,K1,ILL)
C     CALCULATION OF EQUATIONS FOR SIMULATION
      DIMENSION YC(MY,NT,K1),X(MX,NT),A(L,K1),YY(1),YS(1),DY(1),AA(2)
     1,Q(1)
      COMMON RO,RE,CS,CG,RH
      EXTERNAL DIFEQ
      XX=0.0
      HX=X(1,N)/100.0
      DO 10 I=1,L
   10 AA(I)=A(I,K)
      DO 20 I=1,MY
      YY(I)=RO
   20 Q(I)=0.0
      DO 30 I=1,N
      IF(XX-X(1,I)) 40,50,50
   40 DO 60 J=1,MY
   60 YS(J)=YY(J)
      CALL URKGS(YY,DY,XX,HX,MY,AA,L,DIFEQ,Q)
      IF(XX-X(1,I)) 40,50,50
   50 DO 30 J=1,MY
   30 YC(J,I,K)=YY(J)-(YY(J)-YS(J))*(XX-X(1,I))/HX
      ILL=0
      RETURN
      END
```

```
      SUBROUTINE DIFEQ(Y,DY,X,MY,A,L)
C     DRYING-RATE EQUATION (SQUARE, CONSTANT SURFACE)
C     Y(1)= RADIUS OF UNDRYING-CORE
C     X= TIME
C     DY(1)= DIFFERENTIAL VALUE OF Y(1) RESPECT TO X
C     A(1),A(2)= RATE PARAMETERS OF GAS-FILM AND SHELL DIFFUSION
      DIMENSION Y(MY),DY(MY),A(L)
      COMMON R0,RE,CS,CG,RH
      R1=1.0/A(1)
      R2=(R0-Y(1))/((Y(1)/R0)*A(2))
      DY(1)=-R0**2*(CS-CG)/((R1+R2)*RH*Y(1)**2)
      RETURN
      END
      SUBROUTINE GAUYOS(A,N,N1,EPS,ILL)
C     GAUSS-JORDAN METHOD
C     CALCULATION OF SIMULTANEOUS LINEAR EQUATIONS
      DIMENSION A(N,N1)
      DO 10 K=1,N
      BIG=ABS(A(K,K))
      IP=K
      K1=K+1
      IF(K1.GT.N) GO TO 14
      DO 11 I=K1,N
      IF(ABS(A(I,K)).LE.BIG) GO TO 11
      BIG=ABS(A(I,K))
      IP=I
   11 CONTINUE
   14 IF(BIG.GE.EPS) GO TO 12
      ILL=1000
      GO TO 999
   12 IF(IP.EQ.K) GO TO 15
      DO 13 J=1,N1
      TEMP=A(K,J)
      A(K,J)=A(IP,J)
   13 A(IP,J)=TEMP
   15 W=A(K,K)
      DO 20 J=K1,N1
   20 A(K,J)=A(K,J)/W
      DO 30 I=1,N
      IF(I.EQ.K) GO TO 30
      W=A(I,K)
      DO 40 J=K1,N1
   40 A(I,J)=A(I,J)-W*A(K,J)
   30 CONTINUE
   10 CONTINUE
      ILL=0
  999 RETURN
      END
```

```
      SUBROUTINE URKGS(Y,DY,X,HX,MY,A,L,DIFEQ,Q)
C     RUNGE-KUTTA-GILL METHOD
C     CALCULATION OF ORDINARY DIFFERENTIAL EQUATIONS
      DIMENSION Y(MY),DY(MY),A(L),Q(MY),T(20),R(20),P(6),B(4),C(4)
      DATA P(1),B(1),C(1),C(4)/4*0.0/,B(2)/0.2928932/,B(3)/1.707107/
      PX=X
      P(2)=0.5*HX
      P(3)=0.5*HX
      P(6)=0.5*HX
      P(4)=HX
      P(5)=HX
      B(4)=1.0/3.0
      C(2)=0.7071068*HX
      C(3)=-C(2)
      DO 100 J=1,4
      X=PX+P(J)
      CALL DIFEQ(Y,DY,X,MY,A,L)
      DO 100 I=1,MY
      T(I)=P(J+2)*DY(I)-Q(I)
      R(I)=B(J)*T(I)
      Y(I)=Y(I)+R(I)
  100 Q(I)=3.0*R(I)-T(I)+C(I)*DY(I)
      RETURN
      END
```

殻状乾燥モデルに基づく乾燥速度式に関する研究

久保田清・鈴木寛一・保坂秀明・広田令子・伊原　潔

食品の乾燥は，貯蔵とか輸送を容易にすることから，食糧問題を考えていく場合に重要になる単位操作の一つである。各種の乾燥装置を設計し，制御化などを行なっていくためには，乾燥速度式を設定し，それに含まれる速度パラメータを求めていくことが必要である。

一般に食品において，乾燥特性曲線を描いた場合，恒率および減率期間が明確でないが，この原因として食品に特有の表面収縮現象などが考えられる。また，食品は多種にわたる高分子成分によって構成されていて化学的，物理的な性質が単純でなかったり，細胞構造を有しているものが多いことなどからも，水分移動機構などが複雑である。これらの食品に特有な現象を一つずつ究明しながら乾燥速度式を設定していくべきであるが，再現性のよい実験データを得るのが困難であることなどにより容易でない。

本研究は，当面する乾燥装置設計などに対応するために，簡単な乾燥モデルに基づいて半理論的なものでよいから乾燥速度式を設定して，実験データによるシミュレーションを行なっていくことを目的として，つぎに示す3段階で研究を行なった。

(1) 乾燥速度の大きい条件下で適用できると考えられる殻状乾燥モデルを球，長い円柱および薄い平板形について考え，乾燥速度式の設定をした。

(2) 乾燥の進行に伴ってガス境膜拡散律速などから乾燥殻状部拡散律速などに移行していくと考えられる。各形種の乾燥速度式において，何れか一つの律速を仮定して，近似的に乾燥速度パラメータを簡単に求めるための積分式を表面収縮も考慮して誘導した。

(3) 全乾燥期間に適用できる乾燥速度式を得るためには，二つ以上の乾燥速度パラメータを同時に計算していく必要があるが，そのための非線形最小二乗法を使用した電子計算機プログラムを作成した。この計算の乾燥速度パラメータの初期値として (2) で得られる値を使用することができる。

本研究成果は，乾燥機構が複雑な食品などの乾燥速度式を実験データよりシミュレーションによって得る場合に有用である。